

## CHAPTER

## 3

# What Does It Cost? IRR and the Time Value of Money

## CHAPTER CONTENTS

---

 Overview 68
 

---

 3.1. Don't Trust the Quoted Interest Rate—Three Examples 70
 

---

 3.2. Calculating the Cost of a Mortgage 73
 

---

 3.3. Mortgages with Monthly Payments 77
 

---

 3.4. Lease or Purchase? 81
 

---

 3.5. Auto Lease Example 84
 

---

 3.6. More-Than-Once-a-Year Compounding and the EAIR 90
 

---

 3.7. Continuous Compounding and Discounting (Advanced Topic) 93
 

---

 Summing Up 97
 

---

 Exercises 97
 

---

## Overview

In Chapter 2 we introduced the basic tools of financial analysis—present value (PV), net present value (NPV), and internal rate of return (IRR). In Chapters 3–7 we use these tools to answer two basic types of questions:

- *What is it worth?* Presented with an asset—this could be a stock, a bond, a real estate investment, a computer, or a used car—we would like to know *how to value* the asset. The finance tools used to answer this question are mostly related to the concept of present

## CHAPTER 3 What Does It Cost? IRR and the Time Value of Money 69

value (PV) and net present value (NPV). The basic principle is that the value of an asset is the present value of its future cash flows. Comparing this present value to the asset's price tells us whether we should buy it. We introduced PV and NPV in Chapter 2 and we return to them and their applications in Chapter 4.

- *What does it cost?* This sounds like an innocuous question—after all, you usually know the price of the stock, bond, real estate investment, or used car you're trying to value. But many interesting questions of *financing alternatives* depend on the relative interest costs of each alternative. For example, should you pay cash for a car or borrow money to pay for it (and hence make a series of payments over time)? Should you lease that new computer you want or buy it outright? Or perhaps borrow money from the bank to buy it? It's all clearly a question of *cost*—you'd like to pick the alternative that costs the least.

The tools used for the second question—What does it cost?—are mostly derived from the concept of internal rate of return (IRR). This concept—introduced in Chapter 2—measures the compound rate of return of a series of cash flows. In this chapter we'll show you that rate of return, when properly used, can be used to measure the cost of financing alternatives. The main concept presented in this chapter is the *effective annual interest rate* (EAIR), a concept based on the annualized IRR that you can use to compare financing alternatives.

Much of the discussion in this chapter relates to calculating the EAIR and showing its relation to the IRR. We show that the EAIR is a much better gauge of the financing costs than the *annualized percentage rate* (APR), the financing cost often quoted by many lenders such as banks and credit card companies. We show you how to apply this concept to credit-card borrowing, mortgages, and auto leasing. A case that comes with this book applies the concept to student loans.

### Finance Concepts Discussed

- Effective annual interest rate (EAIR)
- Internal rate of return (IRR)
- Annual percentage rate (APR)
- Loan tables
- Mortgage points
- Lease versus purchase

### Excel Functions Used

- IRR
- PMT, IPMT, and PPMT
- Rate
- NPV
- PV
- Exp
- Ln
- Sum
- Goal Seek

### 3.1. Don't Trust the Quoted Interest Rate—Three Examples

To set the stage for the somewhat more complicated examples in the rest of the chapter, we start with three simple examples. Each example shows why *quoted interest rates* are not necessarily representative of costs.

We use the three examples in this section to introduce the concept of EAIR.

The effective annual interest rate (EAIR) is the annualized internal rate of return (IRR) of the cash flows of a particular credit arrangement or security.

#### Example 1: Borrowing from a Bank

In finance “cost” often refers to an interest rate: “I’m taking a loan from the West Hampton Bank because it’s cheaper—West Hampton charges 8% instead of the 9% charged by the East Hampton Bank.” This is a sentence we all understand—8% interest results in lower payments than 9% interest.

But now consider the following alternatives. You want to borrow \$100 for 1 year, and you’ve investigated both the West Hampton Bank and the East Hampton Bank:

- West Hampton Bank is lending at 8% interest. If you borrow \$100 from them today, you’ll have to repay them \$108 in 1 year.
- The East Hampton Bank is willing to lend you any amount you want at a 6% rate. BUT: East Hampton Bank has a “loan initiation charge” of 4%. What this means is that for each \$100 you borrow, you’ll get only \$96, even though you’ll pay interest on the full \$100.<sup>1</sup>

Obviously the cost of West Hampton’s loan is 8%. But is this cheaper or more expensive than the East Hampton loan? You reason as follows: To actually get \$100 in your hands from East Hampton, you’ll have to borrow \$104.17; after they deduct their 4% charge, you’ll be left with \$100 in hand, which is exactly what you need ( $96\% * 104.17 = 100$ ). At the end of a year, you’ll owe East Hampton Bank  $\$104.17 + 6\% \text{ interest} = \$110.42$ . So the actual interest rate

they’re charging you (the effective annual interest rate) is  $EAIR = \frac{110.42}{100} - 1 = 10.42\%$ .

	A	B	C	D
1	<b>CHEAPER LOAN: WEST HAMPTON OR EAST HAMPTON?</b>			
2		<b>West Hampton</b>	<b>East Hampton</b>	
3	Quoted interest rate	8%	6%	
4	Initial charges	0%	4%	
5	Amount borrowed to get \$100 today	100.00	104.17	<-- =100/(1-C4)
6				
7	<b>Date</b>	<b>Cash flow</b>	<b>Cash flow</b>	
8	Date 1, get loan	100.00	100.00	
9	Date 2, pay it back	-108.00	-110.42	<-- =-C5*(1+C3)
10	Effective annual interest rate, EAIR	8.00%	10.42%	<-- =IRR(C8:C9)

<sup>1</sup>Such charges are common in many kinds of bank loans, especially mortgages. They’re obviously a way to increase the cost of the loan and befuddle the customer.

CHAPTER 3 What Does It Cost? IRR and the Time Value of Money 71

This makes everything easier—West Hampton’s 8% loan (EAIR = 8%) is actually cheaper than East Hampton’s “6%” loan (EAIR = 10.42%).

Note in this example that the EAIR is just an IRR, adjusted for the cost taking the loan from East Hampton. EAIR is *always an interest rate*, but usually with some kind of adjustment.

**The lesson of Example 1:** When calculating the cost of financial alternatives, *you must include the fees*, even if the lender (in our case East Hampton Bank) fudges this issue.

**Example 2: Monthly versus Annual Interest**

You want to buy a computer for \$1,000. You don’t have any money, so you’ll have to finance the computer by taking out a loan for \$1,000. You’ve got two financing alternatives:

- Your bank will lend you the money for 15% annual interest. When you ask the bank what this means they assure you that they will give you \$1,000 today and ask you to repay \$1,150 at the end of 1 year.
- Loan Shark Financing Company will also lend you the \$1,000. Their ads say “14.4% annual percentage rate (APR) on a monthly basis.” When you ask them what this means, it turns out that Loan Shark charges 1.2% *per month* (they explain to you that

$$\frac{14.4\%}{12} = 1.2\%.$$

This means that each month Loan Shark adds 1.2% to the loan balance

outstanding at the end of the previous month:

	A	B	C	D	E	F	G	H
1	<b>HOW LOAN SHARK CHARGES:</b>							
2	<b>14.4% PER YEAR ON A MONTHLY BASIS = 1.2% PER MONTH</b>							
3	<b>Loan balance outstanding at the end of each month</b>							
4	Month 0	Month 1	Month 2	Month 3	Month 4	Month 5	Month 6	
5	\$ 1,000.00							
6		\$ 1,012.00						
7			\$ 1,024.14					
8				\$ 1,036.43				
9					\$ 1,048.87			
10						\$ 1,061.46		
11							\$ 1,074.19	
12								
13	Month 7	Month 8	Month 9	Month 10	Month 11	Month 12		
14	\$ 1,087.09	\$ 1,100.13	\$ 1,113.33	\$ 1,126.69	\$ 1,140.21	\$ 1,153.89		
15								
16								
17								
18								
19								

By the end of the year, you will owe Loan Shark \$1,153.89.

$$\$1,153.89 = \$1,500 * (1 + \frac{14.4\%}{12})^{12}$$

Loan Shark’s loan is “compounded monthly.” This means that the 14.4% annual interest translates to 1.2% per month.

Because this is more than the \$1,150 you will owe the bank, you should prefer the bank loan.

72 PART ONE CAPITAL BUDGETING AND VALUATION

The effective annual interest rate (EAIR) of each loan is the *annualized interest rate* charged by the loan. The bank charges you 15% annually and the loan shark charges you 15.39% annually:

	A	B	C	D
1	<b>THE BANK OR LOAN SHARK?</b>			
2		<b>Bank</b>	<b>Loan Shark</b>	
3	Quoted interest rate	15.0%	14.4%	
4	Borrow today	\$ 1,000.00	\$ 1,000.00	
5	Repay in one year	-\$1,150.00	-\$1,153.89	<-- =-C4*(1+C3/12)^12
6	Effective annual interest rate, EAIR	15.00%	15.39%	<-- =IRR(C4:C5)
7				
8	<b>A second way to compute the EAIR</b>			
9	Monthly interest rate		1.20%	<-- =C3/12
10	EAIR		15.39%	<-- =(1+C9)^12-1

Cells C9 and C10 show another way to compute the 15.39% EAIR charged by the loan shark. Cell C9 computes the monthly rate charged by the loan shark as 1.20%, and cell C10

annualizes this rate  $\left(1 + \frac{14.4\%}{12}\right)^{12} - 1 = 15.39\%$ . Thus, there are two ways to compute the EAIR:

$$EAIR = 15.39\% = \left\{ \begin{array}{l} \frac{\text{Payment at end of year}}{\text{Loan taken out beginning of year}} - 1 = \frac{\$1,153.89}{\$1,000.00} - 1 \quad \leftarrow \text{Cell C6} \\ \left(1 + \frac{14.4\%}{12}\right)^{12} - 1 \quad \leftarrow \text{Cell C10} \end{array} \right.$$

**The lesson of Example 2:** The annual percentage rate (APR) does not always correctly reflect the costs of borrowing. To compute the true cost, calculate the effective annual interest rate (EAIR).

**Example 3: An “Interest Free” Loan**

You’re buying a used car. The Junkmobile your heart desires has a price tag of \$2,000. You have two financing options:

- The dealer explains that if you pay cash you’ll get a 15% discount. In this case you’ll pay \$1,700 for the car today. Because you don’t have any money now, you intend to borrow the \$1,700 from your Uncle Frank, who charges 10% interest.
- On the other hand, the dealer will give you “0% financing”: You don’t pay anything now, and you can pay the dealer the full cost of the Junkmobile at the end of the year.

Thus, you have two choices: The dealer’s 0% financing and Uncle Frank’s 10% rate. Which is cheaper?

CHAPTER 3 What Does It Cost? IRR and the Time Value of Money 73

A little thought will show that the dealer is actually charging you an effective annual interest rate (EAIR) of 17.65%: His “0% financing” essentially involves a loan to you of \$1,700 with an end-year repayment of \$2,000:

	A	B	C	D	E
1	<b>FINANCING THE JUNKMOBILE</b>				
2	<b>Year</b>	<b>Pay cash</b>	<b>Dealer's "0% financing"</b>	<b>Differential cash flow</b>	
3	0	-1,700	0	1,700	<-- =C3-B3
4	1		-2,000	-2,000	<-- =C4-B4
5					
6	Effective annual interest rate (EAIR) charged by dealer			17.65%	<-- =IRR(D3:D4)

Uncle Frank’s EAIR is 10%: He will loan you \$1,700 and have you repay only \$1,870. So you’re better off borrowing from him.

**The lesson of Example 3:** Free loans are usually not free! To compute the cost of a “free” loan, calculate the EAIR of the differential cash flows.

### 3.2. Calculating the Cost of a Mortgage

Now that we’ve set the stage, we’ll proceed to a series of somewhat more complicated examples. We start with a mortgage. Housing is most often the largest personal asset an individual owns. Financing housing with a mortgage is something almost every reader of this book will do in his or her lifetime. Calculating the cost of a mortgage is thus a useful exercise. In this chapter mortgages will be one of the examples we use to illustrate the problems encountered in computing the cost of financial assets.

#### A Simple Mortgage

We start with a simple example. Your bank has agreed to give you a \$100,000 mortgage, to be repaid over 10 years at 8% interest. For simplicity, we assume that the payments on the mortgage are annual.<sup>2</sup> The bank calculates the annual payment as \$14,902.95, using Excel’s **PMT** function:

The **PMT** function calculates an annuity payment (a constant periodic payment) that pays off a loan:

$$100,000 = \sum_{t=1}^{10} \frac{14,902.95}{(1.08)^t} = \frac{14,902.95}{(1.08)} + \frac{14,902.95}{(1.08)^2} + \frac{14,902.95}{(1.08)^3} + \dots + \frac{14,902.95}{(1.08)^{10}}$$

<sup>2</sup>In the real world payments are probably monthly; see the example in Section 3.3.

74 PART ONE CAPITAL BUDGETING AND VALUATION

**DIALOG BOX FOR PMT FUNCTION**

The dialog box for Excel's **PMT** function: **Rate** is the interest rate on the loan, **Nper** is the number of repayment periods, and **Pv** is the loan principal. As discussed in Chapter 2, if the loan principal is written as a positive number, Excel presents the payment as a negative number; to avoid this, we write **Pv** as a negative number.

We can summarize all of this in an Excel spreadsheet:

	A	B	C
1	<b>A SIMPLE MORTGAGE</b>		
2	Mortgage principal	100,000	
3	Interest rate	8%	
4	Mortgage term (years)	10	
5	Annual payment	\$14,902.95	<-- =PMT(B3,B4,-B2)
6			
7	<b>Year</b>	<b>Mortgage</b>	
8		<b>cash flow</b>	
8	0	100,000.00	
9	1	-14,902.95	<-- =-\$B\$5
10	2	-14,902.95	
11	3	-14,902.95	
12	4	-14,902.95	
13	5	-14,902.95	
14	6	-14,902.95	
15	7	-14,902.95	
16	8	-14,902.95	
17	9	-14,902.95	
18	10	-14,902.95	
19			
20	Effective annual interest rate (EAIR)	8.00%	<-- =IRR(B8:B18)

CHAPTER 3 What Does It Cost? IRR and the Time Value of Money 75

The EAIR of this particular mortgage is simply the internal rate of return of its payments. Because the payments on the mortgage are annual, the IRR in cell B20 is already in annual terms.

**The Bank Charges “Mortgage Points”**

As in the previous example, you’ve asked the bank for a \$100,000 mortgage. They’ve agreed to give you this mortgage, and they’ve explained that you’ll be asked to repay \$14,902.95 per year for the next 10 years. However, when you get to the bank, you learn that the bank has deducted “1.5 points” from your mortgage. What this means is that you only get \$98,500 (\$100,000 minus 1.5%). Your payments, however, continue to be based on a principal of \$100,000.<sup>3</sup> You realize immediately that this mortgage is more expensive than the mortgage discussed in the previous subsection. The question is: by *how much* is it more expensive? By calculating the EAIR on the mortgage we can answer this question. The calculation below shows that you’re actually paying 8.34% interest annually.

	A	B	C
1	<b>A MORTGAGE WITH POINTS</b>		
2	Mortgage principal	100,000	
3	"Points"	1.50%	
4	Quoted interest	8.00%	
5	Mortgage term (years)	10	
6	Annual payment	\$14,902.95	<-- =PMT(B4,B5,-B2)
7			
8	<b>Year</b>	<b>Mortgage cash flow</b>	
9	0	98,500.00	<-- =B2*(1-B3)
10	1	-14,902.95	<-- =-\$B\$6
11	2	-14,902.95	
12	3	-14,902.95	
13	4	-14,902.95	
14	5	-14,902.95	
15	6	-14,902.95	
16	7	-14,902.95	
17	8	-14,902.95	
18	9	-14,902.95	
19	10	-14,902.95	
20			
21	Effective annual interest rate (EAIR)	8.34%	<-- =IRR(B9:B19)

Note that the EAIR of 8.34% is the IRR of the stream of payments consisting of the actual loan amount (\$98,500) versus the actual payments you’re making (\$14,902.95 annually). Here’s the calculation:

$$98,500 = \sum_{t=1}^{10} \frac{14,902.95}{(1.0834)^t} = \frac{14,902.95}{(1.0834)} + \frac{14,902.95}{(1.0834)^2} + \frac{14,902.95}{(1.0834)^3} + \dots + \frac{14,902.95}{(1.0834)^{10}}$$

<sup>3</sup>Some banks and mortgage brokers also charge an “origination fee,” defined as a payment to cover the initial cost of processing the mortgage. The net effect of “points” and the “origination fee” is the same—you are charged interest on more money than you actually get in hand.



## 76 PART ONE CAPITAL BUDGETING AND VALUATION

At the end of each year, you will report to the Internal Revenue Service the amount of interest paid on the mortgage. Because this interest is an expense for tax purposes, it's important to get it right. To calculate this interest, we need a loan table, which allocates the each year's payment made between interest and repayment of principal (see Section 2.5, page 000). This table is sometimes called an "amortization table" ("amortize" means to repay with a series of periodic payments):

	A	B	C	D	E	F
21	Effective annual interest rate	8.34%	<-- =IRR(B9:B19)			
22						
23	<b>MORTGAGE AMORTIZATION TABLE</b>					
24	Year	Mortgage principal at beginning of year	Payment at end of year	Part of payment that is interest (expense for taxes!)	Part of payment that is repayment of principal (not an expense for tax purposes)	
25	1	98,500.00	\$14,902.95	\$8,211.41	6,691.54	<-- =C25-D25
26	2	91,808.46	\$14,902.95	\$7,653.58	7,249.37	
27	3	84,559.09	\$14,902.95	\$7,049.23	7,853.71	
28	4	76,705.38	\$14,902.95	\$6,394.51	8,508.44	
29	5	68,196.94	\$14,902.95	\$5,685.21	9,217.74	
30	6	58,979.20	\$14,902.95	\$4,916.78	9,986.17	
31	7	48,993.03	\$14,902.95	\$4,084.28	10,818.66	
32	8	38,174.37	\$14,902.95	\$3,182.39	11,720.56	
33	9	26,453.81	\$14,902.95	\$2,205.31	12,697.64	
34	10	13,756.17	\$14,902.95	\$1,146.78	13,756.17	

Column D of the table gives the interest expense for tax purposes. If you report interest payments on your tax return, this is the payment you'd be allowed to report. Note that the interest portion of the annual \$14,902.95 payments gets smaller over the years, whereas the repayment of the principal portion (which is not deductible for tax purposes) gets larger.

### Calculating the Individual Payments with IPMT and PPMT

The above spreadsheet gives the intuition behind the loan table and the split between interest and principal of each payment. The interest and repayment-of-principal payments can be computed directly using the Excel functions **IPMT** and **PPMT**. This is illustrated below.<sup>4</sup>

<sup>4</sup>Note that **IPMT** and **PPMT** work only when the loan payments are equal.

CHAPTER 3 What Does It Cost? IRR and the Time Value of Money 77

	A	B	C	D	E	F	G
	<b>COMPUTING THE MORTGAGE PAYMENTS USING IPMT AND PPMT</b>						
1							
2	Loan principal	100,000					
3	Points	1.50%					
4	Quoted interest	8.00%					
5	Term (years)	10					
6	Effective interest rate	8.34%	=Mortgage with points!B21				
7	Annual payment	\$14,902.95	=PMT(B6,B5,-B2*(1-B3))				
8							
9	<b>MORTGAGE AMORTIZATION TABLE</b>						
	<b>Year</b>	<b>Mortgage principal at beginning of year</b>	<b>Payment at end of year</b>	<b>Part of payment that is interest</b>	<b>Part of payment that is repayment of principal</b>		
10							
11	1	100,000.00	\$14,902.95	\$8,336.46	6,793.44		
12	2	93,206.56	\$14,902.95	\$7,770.13	7,359.77		
13	3	85,846.79	\$14,902.95	\$7,156.58	7,973.31		
14	4	77,873.48	\$14,902.95	\$6,491.89	8,638.01		
15	5	69,235.47	\$14,902.95	\$5,771.79	9,358.11		
16	6	59,877.36	\$14,902.95	\$4,991.65	10,138.24		
17	7	49,739.12	\$14,902.95	\$4,146.48	10,983.42		
18	8	38,755.70	\$14,902.95	\$3,230.85	11,899.04		
19	9	26,856.66	\$14,902.95	\$2,238.89	12,891.00		
20	10	13,965.66	\$14,902.95	\$1,164.24	13,965.66		
21							
22		=IPMT(\$B\$6,A11,\$B\$5,-\$B\$2)					
23							
24				=PPMT(\$B\$6,A11,\$B\$5,-\$B\$2)			

The interest rate from the previous spreadsheet has been formatted to show only two decimal places. The actual interest rate is 8.336459884%

Here is the dialog box for **IPMT** in cell D9 (the syntax of **PPMT** is similar). Note that **Per** specifies the specific period for which the interest is calculated:

**Function Arguments**

IPMT

Rate: \$B\$4 = 0.11

Per: A9 = 1

Nper: \$B\$3 = 10

Pv: -\$B\$2 = -100000

Fv: = number = 11000

Returns the interest payment for a given period for an investment, based on periodic, constant payments and a constant interest rate.

**Rate** is the interest rate per period. For example, use 6%/4 for quarterly payments at 6% APR.

Formula result = \$11,000.00

[Help on this function](#)

OK Cancel

### 3.3. Mortgages with Monthly Payments

We continue with the mortgage examples from Section 3.2. This time we introduce the concept of monthly payments. Suppose you get a \$100,000 mortgage with an 8% interest rate, payable

## 78 PART ONE CAPITAL BUDGETING AND VALUATION

monthly, and suppose you have to pay the mortgage back over 1 year (12 months).<sup>5</sup> Many banks interpret the combination of 8% annual interest and “payable monthly” to mean that the monthly interest on the mortgage is  $\frac{8\%}{12} = 0.667\%$ . This is often referred to as “monthly compounding,” although the usage of this term is not uniform. To compute the monthly repayment on the mortgage, we use Excel’s **PMT** function:

	A	B	C
1	<b>MORTGAGE WITH MONTHLY PAYMENTS</b>		
2	Loan principal	100,000	
3	Loan term (years)	1	
4	Quoted interest rate	8%	
5			
6	<b>Month</b>	<b>Cash flow</b>	
7	0	100,000.00	
8	1	-8,698.84	<-- =PMT(\$B\$4/12,\$B\$3*12,\$B\$2)
9	2	-8,698.84	
10	3	-8,698.84	
11	4	-8,698.84	
12	5	-8,698.84	
13	6	-8,698.84	
14	7	-8,698.84	
15	8	-8,698.84	
16	9	-8,698.84	
17	10	-8,698.84	
18	11	-8,698.84	
19	12	-8,698.84	
20			
21	Monthly IRR	0.667%	<-- =IRR(B7:B19)
22	Effective annual interest rate, EAIR	8.30%	<-- =(1+B21)^12-1

The EAIR on the mortgage in the example is computed by using Excel’s **IRR** function (cell B21). In our case the **IRR** function will give a monthly interest rate of 0.667% (we already

knew this because  $\frac{8\%}{12} = 0.667\%$ ). Annualizing this gives  $8.30\% = \left(1 + \frac{8\%}{12}\right)^{12} - 1$

(cell B22).

### Mortgages: A More Complicated Example

As we saw in Section 3.1, many mortgages in the United States have “origination fees” or “discount points” (the latter are often just called “points”). All of these fees reduce the initial amount given to you by the bank, *without reducing* the principal on which the bank computes its payments (sounds misleading, doesn’t it?).

As an example, consider the above 12-month mortgage with an 8% annual rate, payable monthly, but with an origination fee of 0.5% and 1 point. This means that you actually get

<sup>5</sup>Most mortgages are, of course, for a much longer term. But 12 months enables us to fit the example comfortably within a page. Later we’ll consider longer terms, but the principles will be the same.

## CHAPTER 3 What Does It Cost? IRR and the Time Value of Money 79

\$98,500 (\$100,000 minus \$500 for the origination fee and \$1,000 for the point), but that your monthly repayment remains \$8,698.84:

	A	B	C
1	<b>MORTGAGE EXAMPLE WITH POINTS AND ORIGINATION FEE</b>		
2	Loan principal	100,000.00	
3	Loan term (years)	1	
4	Quoted interest rate	8%	
5	Discount points	1	
6	Origination fee	0.5%	
7			
8	<b>Month</b>	<b>Cash flow</b>	
9	0	98,500.00	<-- =B2*(1-B5/100-B6)
10	1	-8,698.84	<-- =PMT(\$B\$4/12,\$B\$3*12,\$B\$2)
11	2	-8,698.84	
12	3	-8,698.84	
13	4	-8,698.84	
14	5	-8,698.84	
15	6	-8,698.84	
16	7	-8,698.84	
17	8	-8,698.84	
18	9	-8,698.84	
19	10	-8,698.84	
20	11	-8,698.84	
21	12	-8,698.84	
22			
23	Monthly IRR	0.9044%	<-- =IRR(B9:B21)
24	EAIR	11.41%	<-- =(1+B23)^12-1
25			
26	Monthly IRR using Excel's <b>Rate</b> function	0.9044%	<-- =RATE(12,8698.84,-98500)

The monthly IRR (cell B23) is the interest rate that sets the present value of the monthly payments equal to the initial \$98,500 received:

$$\$98,500 = \frac{\$8,698.94}{(1 + 0.9044\%)} + \frac{\$8,698.94}{(1 + 0.9044\%)^2} + \frac{\$8,698.94}{(1 + 0.9044\%)^3} + \dots + \frac{\$8,698.94}{(1 + 0.9044\%)^{12}}$$

$EAIR = 11.41\% = (1 + 0.9044\%)^{12} - 1$  is the annualized cost of the mortgage payments.

As you can see in cell B26, Excel's **Rate** function will also calculate the monthly IRR that we've calculated in cell B23:

80 PART ONE CAPITAL BUDGETING AND VALUATION

**EXCEL NOTE: CALCULATING THE MONTHLY IRR WITH RATE**

The screenshot shows the 'Function Arguments' dialog for the RATE function. The inputs are: Nper = 12, Pmt = 8698.84, Pv = -98500, Fv = (blank), Type = (blank). The result is 0.009044152, which is displayed as 0.9044%. The dialog also includes a description of the function and a 'Formula result' field.

The **Rate** function computes the IRR of a series of constant (the financial jargon is “flat” or “even”) payments so that the discounted value equals the **PV** indicated. Note that in the **Rate** function the signs of the payments (indicated by **Pmt**) and the **Pv** of these payments must be different. This is a feature **Rate** shares with Excel functions like **PMT** and **PV**, discussed in Chapter 2.

**Longer-Term Mortgages**

Suppose the mortgage in the previous example has a 30-year term (meaning:  $360 = 30 * 12$  repayments). Each repayment would be \$733.76 and the EAIR would be 8.4721%:

	A	B	C
	<b>30-YEAR MORTGAGE with points and origination fee</b>		
1			
2	Loan principal	100,000.00	
3	Loan term (years)	30	
4	Quoted interest rate	8%	
5	Discount points	1	
6	Origination fee	0.5%	
7			
8	Initial amount of loan, net of fees	98,500.00	<-- =B2*(1-B5/100-B6)
9	Monthly repayment	733.76	<-- =PMT(B4/12,B3*12,-B2)
10			
11	<b>Calculating the EAIR</b>		
12	Monthly interest rate	0.6800%	<-- =RATE(B3*12,B9,-B8)
13	Effective annual interest rate (EAIR)	8.4721%	<-- =(1+B12)^12-1

We’ve used **PMT** to calculate the payment and **Rate** to compute the monthly interest rate. The EAIR is computed in the usual manner—by compounding the monthly payments (cell B13).

## CHAPTER 3 What Does It Cost? IRR and the Time Value of Money 81

Note that the effect of the initial mortgage fees on mortgage EAIR declines when the mortgage is longer term:

- For the 1-year mortgage discussed previously, the 1.5% initial fee increased the EAIR of the mortgage from 8 to 11.41%.
- For the 30-year mortgage discussed above, the same initial fees increase the EAIR from 8 to 8.4721%.
- The reason that the fees have a smaller effect for the second mortgage is that they are spread out over a much longer term.

### 3.4. Lease or Purchase?

This section uses the concepts of present value and internal rate of return to explore the relative advantages of leasing versus buying an asset. As you will see, the choice between leasing and buying basically comes down to choosing the cheaper of two methods of financing.

Here's our terminology: A *lease* is a rental agreement; in our examples leases will usually be for equipment (we discuss a computer lease and a car lease), but the analysis for real-estate is virtually the same. The party that rents the asset and uses it is called the *lessee* and the owner of the asset is called the *lessor*.

#### A Simple Lease Example

You need a new computer, but you can't decide whether you should buy it or lease it. The computer costs \$4,000. The lessor is your neighborhood computer leasing company, which offers to lease you the computer for \$1,500 per year. The lessor's conditions are that you make four payments of \$1,500: the first payment at the start of the lease (time 0) and subsequent payments at the end of years 1, 2, and 3. Based on past experience, you know that you will keep your new computer for about 3 years. One additional fact: you can borrow from your bank at 15%.

Here is a spreadsheet with the cash flows for the lease and for the purchase:

	A	B	C	D
1	<b>BASIC LEASE VERSUS PURCHASE</b>			
2	Asset cost	4,000.00		
3	Annual lease payment	1,500.00		
4	Bank rate	15%		
5				
6	Year	Purchase cash flow	Lease cash flow	
7	0	4,000.00	1,500.00	
8	1		1,500.00	
9	2		1,500.00	
10	3		1,500.00	
11				
12	PV of costs	4,000.00	4,924.84	<-- =C7+NPV(\$B\$4,C8:C10)
13	Lease or purchase?	purchase		<-- =IF(B12<C12,"purchase","lease")

To decide whether the lease is preferable, we discount the cash flows from both the lease and the purchase at the 15% bank lending rate. We write the outflows as positive numbers, so that the PV in row 12 is the *present value of the costs*. As you can see in cell B12, the PV of the

82 PART ONE CAPITAL BUDGETING AND VALUATION

lease costs is \$4,924.84, which is more than the \$4,000 cost of purchasing the computer. Thus, you prefer the purchase, which is less costly.

There’s another way of doing this same calculation. We compute the IRR of the *differential* cash flows—subtracting the lease cash flow from the purchase cash flow in each of the years:

	A	B	C	D	E
1	<b>BASIC LEASE VERSUS PURCHASE THE DIFFERENTIAL CASH FLOWS</b>				
2	Asset cost	4,000			
3	Annual lease payment	1,500			
4	Bank rate	15%			
5					
6	<b>Year</b>	<b>Purchase cash flow</b>	<b>Lease cash flow</b>	<b>Differential cash flow</b>	
7	0	4,000	1,500	2,500	<-- =B7-C7
8	1		1,500	-1,500	<-- =B8-C8
9	2		1,500	-1,500	<-- =B9-C9
10	3		1,500	-1,500	<-- =B10-C10
11					
12	IRR of differential cash flows			36.31%	<-- =IRR(D7:D10)
13	Lease or purchase?			purchase	<-- =IF(D12>B4,"purchase","lease")
14					
15	Explanation: The lease is like a loan--you save 2,500 in year 0 and pay back 1,500 in each of years 1-3. The IRR of this "loan" is 36.31%.				

The computer lease is equivalent to paying \$1,500 for the computer in year 0 and taking a loan of \$2,500 from the computer leasing company. The computer leasing “loan” has three equal repayments of \$1,500 and an IRR of 36.31%. Because you can borrow money from the bank at 15%, you would prefer to purchase the computer with borrowed money from the bank (at 15%) rather than “borrowing” \$2,500 from the leasing company, which charges 36.31%.

In the spreadsheet below you can see another way to make this same point. If you borrowed \$2,500 from the bank at 15%, you’d have to pay back \$1,094.94 per year for each of the next 3 years (assuming the bank asked for a flat repayment schedule). This is substantially less than the \$1,500 that the computer lessor asks for on the same loan.

$$2,500 = \frac{1,094.94}{1.15} + \frac{1,094.94}{(1.15)^2} + \frac{1,094.94}{(1.15)^3}$$

The conclusion is that if you decide to borrow \$2,500 to purchase the computer, you should do so from the bank rather than from the computer leasing company. In the spreadsheet, we’ve used the Excel **PMT** function to compute the repayment:

	A	B	C	D	E
17	<b>What if you borrowed \$2,500 from the bank?</b>				
18	<b>Year</b>	<b>Money saved by leasing</b>		<b>Same amount from bank</b>	
19	0	2,500		2,500.00	
20	1	-1,500		-1,094.94	<-- =PMT(\$B\$4,3,\$D\$19)
21	2	-1,500		-1,094.94	
22	3	-1,500		-1,094.94	

### What Have We Assumed about Leasing versus Purchasing?

The leasing example we’ve considered above illustrates the spirit of lease/purchase analysis. The example makes some simplifying assumptions that are worth noting:

## CHAPTER 3 What Does It Cost? IRR and the Time Value of Money 83

- No taxes: When corporations lease equipment, the lease payments are expenses for tax purposes; when these corporations buy assets, the depreciation on the asset is an expense for tax payments. Taxes complicate the analysis somewhat; the case of leasing with taxes is considered in Chapter 5.
- Operational equivalence of lease and purchase: In our analysis we don't ask whether you need a computer—we assume that you've already answered this question positively, so that only the method of acquisition is in question. Our analysis also assumes that any maintenance or repairs that need to be done on the computer will be done by you, whether you lease or buy the computer.
- No residual value: We've assumed that the asset (in this case, the computer) is worthless at the end of the lease term.

We explore the last point briefly. Suppose you think that the computer will be worth \$800 at the end of year 3. Then—as shown below—the purchase cash flows change, so that owning the computer gives you an inflow of \$800 in year 3.<sup>6</sup> The cost of purchasing the computer is reduced (the present value of the purchase is now \$3,304) and the lease alternative becomes even less attractive than the purchase. Another way of seeing this is to look at the IRR of the differential cash flows, which is now 45.07% (cell B23).<sup>7</sup>

	A	B	C	D
1	<b>LEASE VERSUS PURCHASE WITH RESIDUAL VALUE</b>			
2	Asset cost	4,000.00		
3	Annual lease payment	1,500.00		
4	Residual value, yr. 3	800	<-- Value of computer at end year 3	
5	Bank rate	15%		
6				
7	<b>Year</b>	<b>Purchase cash flow</b>	<b>Lease cash flow</b>	
8	0	4,000.00	1,500.00	
9	1		1,500.00	
10	2		1,500.00	
11	3	-800.00	1,500.00	
12				
13	PV of costs	3,304.35	4,924.84	<-- =C8+NPV(\$B\$5,C9:C11)
14	Lease or purchase?		purchase	<-- =IF(B13<C13,"purchase","lease")
15				
16	<b>Calculating the IRR of the differential cash flow</b>			
17	<b>Year</b>		<b>Money saved by leasing</b>	
18	0		2,500.00	<-- =B8-C8
19	1		-1,500.00	<-- =B9-C9
20	2		-1,500.00	
21	3		-2,300.00	
22				
23	IRR of differential		45.07%	<-- =IRR(C18:C21)
24	Lease or purchase?		purchase	<-- =IF(C23>B5,"purchase","lease")

<sup>6</sup>Note that because we're writing *outflows* (like the cost of the computer) as *positive numbers*, we have to write the inflows as negative numbers.

<sup>7</sup>A caveat is in order here: We're treating the computer's residual value as if it has the same certainty as the rest of the cash flows, whereas clearly it is less certain. The finance literature has a technical solution to this: We find the *certainty equivalent* of the residual value. For example, it may be that we expect the residual value to be \$1200, but that—recognizing the uncertainty of getting this value—we treat this as equivalent to getting an \$800 residual with certainty.



## 84 PART ONE CAPITAL BUDGETING AND VALUATION

## 3.5. Auto Lease Example

Here's a slightly more realistic (and more complicated) example of leasing: You've decided to get a new car. You can either lease the car or buy it; if you decide to buy the car, you can finance with a 3% bank loan. The relevant facts are given in the spreadsheet below, but we'll summarize them here:

- The manufacturer's suggested retail price (MSRP) for the car is \$24,550, but you've been able to negotiate a price of \$22,490 with the dealer.<sup>8</sup> In the jargon of the car leasing business, the \$22,490 is referred to as the "capitalized cost." To this price must be added a destination charge of \$415, so that you end up paying \$22,905 if you purchase the car. This price represents your alternative purchase cost if you decide to buy instead of lease the car.
- The dealer has offered you the following lease terms:
  - You pay \$1,315 at the signing of the lease. The dealer explains that this is the total of \$415 "destination charge," \$450 "acquisition fee," and a \$450 security deposit. The security deposit will be refunded at the end of the lease.
  - You will pay \$373.43 per month for the next 24 months. In month 24 you get your security deposit of \$450 back.
  - You guarantee that the car will have a residual value of \$13,994 at the end of the lease. The dealer has based this value on 57% of the MSRP. What this means is that if the car is worth less than \$13,994 at the end of the 24th month, the lessee (you) will make up the difference.<sup>9</sup> The end-lease payment associated with this residual can be written as:

$$\text{End-lease residual payment} = \begin{cases} 13,994 - \text{market value} & \text{if market value} < 13,994 \\ 0 & \text{otherwise} \end{cases}$$

Another way of writing this payment is  $\text{Max}(13,994 - \text{market value}, 0)$ . The  $\text{max}(A, B)$  notation means that you pay the larger of  $A$  or  $B$ . Conveniently, **Max** is also a function in Excel.

The residual value turns out to be an important factor in the way you view leasing versus purchase. We'll devote more time to it later. For the moment, let's assume that you think the car will actually be worth \$15,000 at the end of 2 years so your last payment on the lease is zero:

$$\begin{aligned} \text{End-lease residual payment} &= \text{Max}(13,994 - \text{market value}, 0) \\ &= \text{Max}(13,994 - 15,000, 0) = \text{Max}(-1,006, 0) = 0 \end{aligned}$$

All of the lease costs are listed in column C of the spreadsheet below. To evaluate these costs, look at column D, which shows the costs associated with buying the car; there are only

<sup>8</sup>The "manufacturer's suggested retail price" (MSRP—also referred to as the car's "sticker price"—is the price the auto manufacturer suggests as an appropriate price for the car. In reality it's a kind of official fiction and forms the basis for negotiation between the dealer and the car purchaser. In our example the MSRP is used in the residual value computation, but the actual price paid for the car is less.

<sup>9</sup>According to [www.edmunds.com](http://www.edmunds.com): "The lease-end fees are generally reasonable, unless the car has 100,000 miles on it, a busted-up grille and melted chocolate smeared into the upholstery. Dealers and financial institutions want you to buy or lease another car from them, and can be rather lenient regarding excess mileage and abnormal wear. After all, if they hit you with a bunch of trumped-up charges you're not going to remain a loyal customer, are you? . . . But keep in mind that if you take your business elsewhere, you're going to be facing a bill for items like worn tires, paint chips, door dings, and the like."

CHAPTER 3 What Does It Cost? IRR and the Time Value of Money 85

two: the initial purchase price of the car (\$22,490 + the destination cost of \$415 = \$22,905) and what you anticipate will be the market value of the car at the end of the lease term (in the example below, you think the car will actually be worth \$15,000). Because we've used the convention of making costs positive numbers, the inflow from selling the car is a negative number.

This last number bears some examination: If you lease, your last payment is

$$\begin{aligned} \text{last lease payment} &= \text{last month's rental} - \text{return of security deposit} \\ &\quad + \text{end-of-lease residual payment} \\ &= 373.43 - 450 + \text{Max}(13,994 - \text{market value}, 0) \end{aligned}$$

If you're right, and the actual market value of the car is \$15,000, then your last "payment" is -\$76.57 (meaning that you'll get \$76.57 back from the lease company).

	A	B	C	D	E	F
1	<b>AUTO LEASE VERSUS PURCHASE</b>					
2	MSRP	24,550	<-- Manufacturer's suggested retail price			
3	Capitalized cost	22,490	<-- Negotiated price			
4	Destination charge	415	<-- Paid both by the lessee and the buyer			
5	Acquisition fee	450	<-- Paid only by the lessee			
6	Security deposit	450	<-- refunded at end of lease			
7						
8	Payment due at signing	1,315	<-- =SUM(B4:B6)			
9	Monthly payment	373.43	<-- Dealer's lease offer			
10						
11						
12	Residual value after 2 years as % of MSRP	57%				
13	Lease residual value after 2 years	13,994	<-- =B12*B2--lessee guarantees this value			
14	Your estimated residual value	15,000	<-- Your guess			
15						
16						=B3+B4
17						
18		Month	Payment	Purchase	Difference	
19		0	1,315.00	22,905.00	21,590.00	<-- =D18-C18
20		1	373.43		-373.43	<-- =D19-C19
21		2	373.43		-373.43	<-- =D20-C20
22		3	373.43		-373.43	
23		4	373.43		-373.43	
24		5	373.43		-373.43	
25		6	373.43		-373.43	
26		7	373.43		-373.43	
27		8	373.43		-373.43	
28		9	373.43		-373.43	
29		10	373.43		-373.43	
30		11	373.43		-373.43	
31		12	373.43		-373.43	
32		13	373.43		-373.43	
33		14	373.43		-373.43	
34		15	373.43		-373.43	
35		16	373.43		-373.43	
36		17	373.43		-373.43	
37		18	373.43		-373.43	
38		19	373.43		-373.43	
39		20	373.43		-373.43	
40		21	373.43		-373.43	
41		22	373.43		-373.43	
42		23	373.43		-373.43	
43		24	-76.57	-15,000.00	-14,923.43	<-- =D42-C42
44						
45						
46						
47						
48						
49						
			Monthly IRR		0.44%	<-- =IRR(E18:E42)
			EAIR		5.39%	<-- =(1+E44)^12-1
			Buy or lease?			
			Alternative financing		7%	
			Buy or lease?		lease	<-- =IF(E48>E45,"lease","buy")

## 86 PART ONE CAPITAL BUDGETING AND VALUATION

Column E in the spreadsheet subtracts the lease from the purchase cash flows. Initially, the lease saves you \$21,590; in months 1–23, the lease costs you \$373.43 more than the purchase, and at the end of month 24 the lease costs you \$14,923.43 more than the purchase.


The monthly IRR of the differential cash flows is 0.44%, which gives an EAIR of 14.36% (cells E44 and E45).

Should you buy or should you lease? It depends on your alternative cost of financing. If you can finance at a bank for less than 5.39%, then you should buy the car; otherwise, the lease looks like a good deal. In our case you can finance at the bank for 3% (cell B48), so you should buy the car with a bank loan instead of leasing it.

### The Role of the Residual

The residual value of the car is very important in determining the cost of the lease. To illustrate this we use the **Data|Data Tools|What-if Analysis|Data Table** feature of Excel (see Chapter 28) to run a sensitivity table that shows the EAIR and the lease/buy decision as a function of your estimated end-lease market value of the car:

	A	B	C	D	E	F	G
52	<b>Data table: The EAIR of the lease</b> as a function of the market value of car at end of lease						
53	<b>Estimated market value of car at end-lease</b>	<b>EAIR</b>	<b>Lease or buy?</b>				
54				← Data table header has been hidden			
55	10,000	2.6%	lease				
56	10,500	2.6%	lease				
57	11,000	2.6%	lease				
58	11,500	2.6%	lease				
59	12,000	2.6%	lease				
60	12,500	2.6%	lease				
61	13,000	2.6%	lease				
62	13,500	2.6%	lease				
63	14,000	2.6%	lease				
64	14,500	4.0%	lease				
65	15,000	5.4%	lease				
66	15,500	6.7%	lease				
67	16,000	8.1%	buy				
68	16,500	9.4%	buy				
69	17,000	10.7%	buy				
70	17,500	11.9%	buy				
71	18,000	13.2%	buy				
72							



As the data table shows, leasing is preferable if you think that the actual market value at the end of the lease term will be low relative to the lease residual of \$13,994. The lease is based on your “reselling” the car to the dealer for \$13,994; if you think that the actual market value of the car will be much higher, then you would be selling it to the dealer at a loss, and you’re better off buying the car and reselling it yourself.<sup>10</sup> The break-even market value—the estimated market value for which you are indifferent between leasing the car or financing it at the bank at 3%—is

<sup>10</sup>Some leases actually give you the option of buying the car for the residual value at the end of the lease term. This effectively locks in the lease EAIR, because if the car is worth more than the lease residual value, you can always buy it for the residual and resell the car on the open market.

## CHAPTER 3 What Does It Cost? IRR and the Time Value of Money 87

somewhere between \$14,000 and \$14,500; in this range the EAIR of the lease is 3%, which is equal to the cost of the alternative financing.<sup>11</sup>

### Financing the Purchase of a Car with a Bank Loan: Cheaper or More Expensive?

In our analysis above we concluded that the annual cost of the 2-year lease (cell E45) is an EAIR of 5.39%. So it stands to reason that if you can get a cheaper loan from a bank, you should to take the bank loan and use the proceeds to buy the car. And yet... suppose the bank offers you a 3% loan (with monthly compounding, so that the monthly interest rate is  $3\%/12 = 0.25\%$ ), and suppose that we have the same amount to finance (namely, the car cost of \$22,905 minus the money down of \$1,315). The spreadsheet below shows that the monthly payments on this bank loan are much *larger* than those of the car lease:

	H	I	J
2	<b>Financing with a bank loan</b>		
3	Cost of car	22,905	<-- =B3+B4
4	Money down	1,315	
5	Amount to finance	21,590	<-- =I3-I4
6			
7	Bank rate	3%	
8	Monthly loan payment	927.96	<-- =PMT(I7/12,24,I5)
9	Monthly lease payment	373.43	
10			
11	<b>Note:</b> The PMT function in cell H8 calculates the		
12	monthly payment on a \$21,590 loan (cell H5) given for 24		
13	months at monthly interest 0.25% (cell H7/12).		

Now this is confusing: Borrowing from the bank at 3% to buy the car involves a much higher monthly payment (\$926.96) than the monthly lease payment (\$373.43). Yet in our first analysis of the lease on page 000, we concluded that a loan rate of 3% is preferable to the lease EAIR of 5.39%. To resolve this apparent contradiction, recall that the difference between the two—the lease and the loan—is the residual value built into the lease: This residual value—essentially a guarantee that you, the lessee, extend to the auto lessor—both reduces your monthly lease payments *and* increases your stake in the residual value of the car. Compared with the bank loan, the lease gives you lower payments in return for bearing the higher risk of guaranteeing the residual value of the car. There is no free lunch.

To see that the loan is actually cheaper, assume that you take a separate bank loan to finance the car's residual value of \$15,000 in 2 years:

$$PV \text{ of residual value} = \frac{\$15,000}{\left(1 + \frac{3\%}{12}\right)^{24}} = \$14,127.53.$$

<sup>11</sup>The exact residual value for which you are indifferent is \$14,134.

88 PART ONE CAPITAL BUDGETING AND VALUATION

We can now divide the purchase price of \$22,905 into two parts:

$$\$22,905 = \underbrace{\$14,127.53}_{\frac{\$15,000}{\left(1 + \frac{3\%}{12}\right)^{24}}} + \$8,777.47$$

The total cost of \$8,777.47 is the cost of using the car for the next 2 years. Of this amount, you have to pay an immediate down payment of \$1,315, which leaves \$7,462.47 to finance. Financing this amount with a lease will cost \$373.43 per month, whereas financing with a bank loan will cost \$320.75 per month:

	A	B	C
	<b>AUTO LEASE VERSUS PURCHASE COMPARING BANK LOAN TO LEASE PAYMENT</b>		
1			
2	MSRP	24,550.00	<-- Manufacturer's suggested retail price
3	Capitalized cost	22,490.00	<-- Negotiated price
4	Destination charge	415.00	<-- Paid both by the lessee and the buyer
5	Acquisition fee	450.00	<-- Paid only by the lessee
6	Security deposit	450.00	<-- refunded at end of lease
7			
8	Payment due at signing	1,315.00	<-- =SUM(B4:B6)
9	Monthly payment	373.43	<-- Dealer's lease offer
10			
11	Residual value after 2 years as % of MSRP	57%	
12	Lease residual value after 3 years	13,993.50	<-- =B11*B2--lessee guarantees this value
13	Your estimated residual value	15,000.00	<-- Your guess
14			
15	<b>Financing with a bank loan</b>		
16	Bank rate	3%	
17	Monthly rate	0.25%	<-- =B16/12
18			
19	Cost of car	22,905.00	<-- =B3+B4
20	Money down	1,315.00	
21	Amount to finance	21,590.00	<-- =B19-B20
22			
23	Loan principal to finance residual in 2 years	14,127.53	<-- =B13/(1+B17)^24
24	Loan principal to finance car lease for 2 years	7,462.47	<-- =B19-B20-B23
25			
26	Monthly loan payment to finance car lease for 2 years	320.75	<-- =PMT(B17,24,-B24)
27	Monthly lease payment	373.43	

The bank loan is cheaper. We've summarized our logic in Figure 3.1.




FIGURE 3.1 Car lease versus car loan.

### 3.6. More-Than-Once-a-Year Compounding and the EAIR

Suppose you are charged interest on a monthly basis, but you want to compute the annual interest cost. Here's an example: XYZ Bank says that it charges an annual percentage rate (APR) of 18% on your credit card balances, with "interest computed monthly." Suppose that what the bank means is that it charges 1.5% per month on the outstanding balance at the beginning of the month. To determine what this means in practice, you should ask yourself "If I have a credit balance of \$100 outstanding for 12 months, how much will I owe at the end of the 12-month period?" If we set this up in Excel, we get the following:

	A	B	C	D
1	<b>MONTHLY COMPOUNDING OF CREDIT CARD BALANCES</b>			
2	"Annual" rate	18%		
3	Monthly rate	1.5%	<-- =B2/12	
4				
5	<b>Month</b>	<b>Balance at beginning of month</b>	<b>Interest for month</b>	<b>Balance at end of month</b>
6	1	100.00	1.50	101.50
7	2	101.50	1.52	103.02
8	3	103.02	1.55	104.57
9	4	104.57	1.57	106.14
10	5	106.14	1.59	107.73
11	6	107.73	1.62	109.34
12	7	109.34	1.64	110.98
13	8	110.98	1.66	112.65
14	9	112.65	1.69	114.34
15	10	114.34	1.72	116.05
16	11	116.05	1.74	117.79
17	12	117.79	1.77	119.56
18				
19	Effective annual interest rate (EAIR)	19.56%	<-- =D17/B6-1	
20		19.56%	<-- =(1+B3)^12-1	

At the end of 12 months you would owe \$119.56—the initial \$100 balance plus \$19.56 in interest. Cells B19 and B20 show two ways of calculating the effective annual interest rate:

- In cell B19, we take the end-year balance that results from the initial \$100 credit card balance and divide it by the initial balance to calculate the interest rate:

$$\begin{aligned}
 EAIR &= \frac{\text{End-year balance}}{\text{Initial balance}} - 1 \\
 &= \frac{119.56}{100} - 1 = 19.56\%
 \end{aligned}$$

- In cell B20 we take the monthly interest rate and compound it:

$$\begin{aligned}
 EAIR &= (1 + \text{monthly rate})^{12} - 1 \\
 &= (1.015)^{12} - 1 = 19.56\%
 \end{aligned}$$

When the annual interest rate  $r$  is compounded  $n$  times per year, the  $EAIR = \left(1 + \frac{r}{n}\right)^n - 1$ .

### APR and EAIR

By an act of Congress (“The Federal Truth in Lending Act”) lenders are required to specify the annual percentage rate (APR) charged on loans. Unfortunately, the Truth in Lending Act does not specify how the APR is to be computed, and the use of the term by lenders is not uniform. Although “APR” is legal terminology designed to help the consumer understand the true cost of borrowing, in practice the APR is not well-defined and may not represent the actual cost of borrowing. Sometimes the APR is the effective annual interest rate (EAIR), but in other cases—like the credit card example in this section—the APR is something else. The result is much convoluted wording and a lot of confusion.<sup>12</sup>

### The EAIR and the Number of Compounding Periods per Year $n$

In the above example, the credit card company takes its 18% “annual” interest rate charge and turns it into a 1.5% monthly interest rate. As we saw, the resulting EAIR is 19.56%.

In Figure 3.2 we compute the effect of the number of compounding periods on the EAIR:

THE EFFECTIVE ANNUAL INTEREST RATE (EAIR) AND THE NUMBER OF COMPOUNDING PERIODS		
The stated annual interest rate is 18%		
Number of compounding periods per year	EAIR formula	EAIR (%)
1	$(1 + 18\%) - 1$	18.00
2 (Semiannual compounding)	$\left(1 + \frac{18\%}{2}\right)^2 - 1$	18.81
4 (Quarterly compounding)	$\left(1 + \frac{18\%}{4}\right)^4 - 1$	19.252
12 (Monthly compounding)	$\left(1 + \frac{18\%}{12}\right)^{12} - 1$	19.562

<sup>12</sup>A case that accompanies this book gives three actual APR examples and the resulting EAIR. In each case the definition of APR used by the lender is different. In only one of the three cases does the APR correspond to the EAIR.



## 92 PART ONE CAPITAL BUDGETING AND VALUATION

24 (Semimonthly compounding)	$\left(1 + \frac{18\%}{24}\right)^{24} - 1$	19.641
52 (Weekly compounding)	$\left(1 + \frac{18\%}{52}\right)^{52} - 1$	19.685
365 (Daily compounding)	$\left(1 + \frac{18\%}{365}\right)^{365} - 1$	19.716

FIGURE 3.2: The EAIR when an annual interest rate of 18% is compounded for various times per year.

The EAIR grows with the number of compounding periods. The EAIR is

$$EAIR = \left(1 + \frac{\text{stated annual interest rate}}{\text{number of annual compounding periods}}\right)^{\text{number of compounding periods per year}} - 1$$

When we do this in Excel, we see that the EAIR grows as the number of compounding periods increases. For a very large number of compounding periods, the EAIR approaches a limit of 19.722% (cell C20 below):

There are two important things to note about the EAIR computation:

- As the number of compounding periods per year  $n$  increases, the  $EAIR = \left(1 + \frac{r}{n}\right)^n - 1$  gets higher.
- The rate at which the EAIR increases gets smaller as the number of annual compounding periods gets larger. There is very little difference between the EAIR when interest is compounded 36 times per year (EAIR = 19.668%) and the EAIR when we compound 365 times per year (EAIR = 19.716%).

AQ: Please confirm if this placement ok.

	A	B	C	D	E
1	<b>EAIR AND NUMBER OF COMPOUNDING PERIODS</b>				
2	"Annual" rate	18%			
3					
4	<b>Number of compounding periods per year</b>	<b>Rate per period</b>	<b>EAIR</b>		
5	1	18.00%	18.000%	<-- =(1+B5)^A5-1	
6	2	9.00%	18.810%	<-- =(1+B6)^A6-1	
7	3	6.00%	19.102%	<-- =(1+B7)^A7-1	
8	4	4.50%	19.252%		
9	6	3.00%	19.405%		
10	12	1.50%	19.562%		
11	24	0.75%	19.641%		
12	36	0.50%	19.668%		
13	52	0.35%	19.685%		
14	100	0.18%	19.702%		
15	150	0.12%	19.709%		
16	200	0.09%	19.712%		
17	250	0.07%	19.714%		
18	300	0.06%	19.715%		
19	365	0.05%	19.716%		
20	infinite		19.722%	<-- =EXP(B2)-1	
21					
22					
23					
24					
25					
26					
27					
28					
29					
30					
31					
32					
33					
34					
35					
36					
37					
38					

### 3.7. Continuous Compounding and Discounting (Advanced Topic)

In cell C20 we compute the limit of the EAIR when the number of compounding periods gets very large. This limit is called *continuous compounding*. For  $n$  annual compounding periods per year, the  $EAIR = \left(1 + \frac{r}{n}\right)^n - 1$ . When the number of annual compounding periods  $n$  gets

very large, the EAIR becomes close to  $e^r - 1$ . The number  $e = 2.71828182845904$  is the base of natural logarithms and is included in Excel as the function **Exp( )**. In the jargon of finance,  $e^{rt}$  is called the *continuously compounded future value after  $T$  years at annual interest rate  $r$* . In the spreadsheet below you can see the difference between the *discretely compounded* future value and the *continuously compounded* future value:


94 PART ONE CAPITAL BUDGETING AND VALUATION

	A	B	C
1	<b>CONTINUOUS COMPOUNDING</b>		
2	"Annual" rate	18%	
3	Number of compounding periods per year	250	
4	Number of years, T	3	
5	Effective annual interest rate, EAIR		
6		19.71%	
7	Discretely compounded future value after t years $= (1+EAIR)^T$	1.7157	$\leftarrow = (1+B2/B3)^{(B3*B4)}$
8	Continuously compounded future value $= e^{rT}$	1.7160	$\leftarrow = EXP(B2*B4)$


When the number of compounding periods gets very large, the difference between the discrete and continuous interest rate becomes very small.

### The Continuously Compounded Discount Factor

In Chapter 2 we saw that future value and present value are closely related:



A similar relation holds for continuous compounding:



The following spreadsheet summarizes these relations:

	A	B	C
1	<b>DISCRETE AND CONTINUOUS COMPOUNDING</b>		
2	Interest rate	10%	
3	Initial amount, X	100	
4	Terminal date, T	3	
5	Discretely compounded future value, $X*(1+r)^T$	133.100	$\leftarrow = B3*(1+B2)^{B4}$
6	Continuously compounded future value, $X*e^{rT}$	134.986	$\leftarrow = B3*EXP(B2*B4)$
7			
8	Interest rate	10%	
9	Terminal amount, Y	100	
10	Terminal date, T	3	
11	Discretely compounded present value, $Y/(1+r)^T$	75.131	$\leftarrow = B9/(1+B8)^{B10}$
12	Continuously compounded present value, $Y*e^{-rT}$	74.082	$\leftarrow = B9*EXP(-B10*B8)$

### An Actual Credit Card Example

Continuously compounded interest may seem like an ethereal concept—highly theoretical but not very useful. The example in this subsection shows how useful continuously compounded interest can actually be. The Columbus State University credit card in the ad below charges a penalty annual percentage rate (APR) of 27.99%.<sup>13</sup> The parentheses in the ad make it clear that the company is actually charging 0.07669% *per day* on outstanding balances. This rate is calculated by taking 27.99% and dividing it by the number of days per year:

$$0.07669\% = \frac{27.99\%}{365}$$

<b>Annual Percentage Rate (APR) for Purchases</b>	10.99 % (0.03011% Daily Periodic Rate (DPR))
<b>Other APRs</b>	<b>Balance Transfer APR:</b> 0% for six months (0.000000% DPR), then the standard APR for purchases <b>Cash Advance APR:</b> 14.99% (0.04107% DPR) <b>Penalty APR:</b> 27.99% (0.07669% DPR) <sup>1</sup>
<b>Variable Rate Information</b>	Your APR for purchases (including balance transfer) and cash advance transactions will vary. The rate is determined monthly by

If you carried a \$100 balance throughout the year, you would owe  $100 \cdot (1.007668)^{365}$  at the end of the year.<sup>14</sup> As the spreadsheet shows, this translates into a 32.286% EAIR (cell B5):

<sup>13</sup>What is the penalty rate? As the card Web site explains: “If you are late making a payment, any rates not exceeding the Penalty Rate may change to the same rate and type as the Penalty Rate.” In other words, if you are overdue on *any* payment, the penalty rate applies to all existing balances.

<sup>14</sup>Note the small discrepancy between our Excel and the Visa advertisement:  $27.99\% / 365 = 0.0766849\%$ . By the normal rules of rounding off, this should be rounded down to 0.07668%, but Columbus State’s Visa takes the extra 0.0001%! As they say: “Every little bit helps.”

## 96 PART ONE CAPITAL BUDGETING AND VALUATION

	A	B	C
1	<b>COLUMBUS STATE VISA CARD</b>		
2	APR	27.99%	
3	Daily	0.07668%	<-- =B2/365
4			
5	Effective annual interest rate (EAIR)	32.286%	<-- =(1+B3)^365-1
6	Continuously compounded interest	32.300%	<-- =EXP(B2)-1

As you can see in cell B6, essentially the same interest rate can be computed using continuous compounding. From a computational point of view, using continuous compounding is simpler than discretely compounding a daily interest rate.

### CONTINUOUS COMPOUNDING IN THIS BOOK

We rarely use continuous compounding in this book, except in the options chapters (Chapters 20–23). In other cases we use only discrete compounding, although we may occasionally point out in a footnote the continuously compounded counterpart of a discretely compounded calculation.

### Computing the Continuous Return

In past subsections we have measured discrete versus continuous compounding. In this section we show how to measure discrete versus continuous rates of return. Suppose that an investment grows from  $X$  at time zero to  $Y$  at time  $T$ . Then the *discretely compounded annual return* on the investment is  $r = \left(\frac{Y}{X}\right)^{1/T} - 1$ . The *continuously compounded annual return* is given by  $r = \frac{1}{T} \text{Ln}\left(\frac{Y}{X}\right)$ . In the example below, an investment of \$100 grows to \$200 over a period of 4 years.<sup>15</sup> The spreadsheet computes both the discretely compounded and the continuously compounded annual returns:

	A	B	C
1	<b>COMPUTING DISCRETE AND CONTINUOUS RETURNS</b>		
2	Investment at time 0	100	
3	Value at time T	200	
4	T (years)	4	
5			
6	Discretely compounded annual return	18.92%	<-- =(B3/B2)^(1/B4)-1
7	Continuously compounded annual return	17.33%	<-- =(1/B4)*LN(B3/B2)
8			
9	<b>Why is this true: Compounding the initial investment to get the future value</b>		
10	Discrete compounding	200	<-- =B2*(1+B6)^B4
11	Continuous compounding	200	<-- =B2*EXP(B7*B4)

<sup>15</sup>The Excel function **Ln** computes the natural logarithm of a number. If  $\text{Ln}(a) = b$ , then  $e^b = a$ .

## CHAPTER 3 What Does It Cost? IRR and the Time Value of Money 97

The discretely compounded annual return (cell B6) is 18.92%; this is proved in cell B10, where we show that  $200 = 100 * (1 + 18.92\%)^4$ . In other words, over 4 years the discretely compounded future value of \$100 at 18.92% annually is \$200.

The continuously compounded annual return is 17.32% (cell B7); this is proved in cell B11, where we show that  $200 = 100 * \exp(17.32\% * 4)$ . Over 4 years the continuously compounded future value of \$100 at 17.32% annually is \$200.

Note that you would be indifferent in choosing between an annual discretely compounded return of 18.92% and an annual continuously compounded return of 17.32%. Over any given time frame, both rates make an initial investment grow to the same final result!

## Summing Up

In this chapter we've applied the time value of money (PV, NPV, and IRR) to a number of relevant problems:

- Finding the effective annual interest rate (EAIR): This is the compound annual interest rate implicit in a specific financial asset; another way to think about this is that it's the annualized IRR. We've given a number of examples—leases, mortgages, credit cards—all of which illustrate that the only way to evaluate the financing cost is by calculating the EAIR.
- The effect of nonannual compounding periods: Many interest rates are calculated on a monthly or even a daily basis. The EAIR demands that we annualize these interest rates so that we can compare them. When the number of compounding periods gets very large (like our Columbus State University example), the  $EAIR = er$ , where  $e = 2.71828182845905$  (computed by `=Exp()` in Excel) and  $r$  is the stated interest rate.

## Exercises

1. You are considering buying the latest stereo system model. The dealer in “The Stereo World” store has offered you two payment options. You can either pay \$10,000 now, or you can take advantage of their special deal and “buy now and pay a year from today,” in which case you will pay \$11,100 in 1 year. Calculate the effective annual interest rate (EAIR) of the store's special deal.
2. You have two options of paying for your new dishwasher: You can either make a single payment of \$400 today, or you can pay \$70 for the next 6 months, with the first payment made today. What is the effective annual interest rate (EAIR) of the second option?
3. Your lovely wife has decided to buy you a vacuum cleaner for your birthday (she always supports you in your hobbies...). She called your best friend, a manager of a vacuum cleaner store, and he has suggested one of two payment plans: She can either pay \$100 now or make 12 monthly payments of \$10 each, starting from today. What are the monthly IRR and the EAIR of the payment-over-time plan?
4. Your local bank has offered you a mortgage of \$100,000. There are no points, no origination fees, and no extra initial costs (meaning you get the full \$100,000). The mortgage is to be paid back over 10 years in annual payments, and the bank charges 12% annual interest.
  - a. Calculate the monthly mortgage payments
  - b. What is the mortgage EAIR?

## 98 PART ONE CAPITAL BUDGETING AND VALUATION

5. Your local bank has offered you a 20-year, \$100,000 mortgage. The bank is charging 1.5 points and “processing” costs of \$750; both points and processing costs are deducted from the mortgage when it is given. Payments on the mortgage are annual and are based on a 10% interest rate on the full amount of the mortgage (that is, \$100,000).
  - a. Calculate the annual mortgage payment
  - b. Calculate the EAIR
  - c. Compute an amortization table that shows the amount of interest you can report for taxes each year.
6. Your local bank has offered you a 10-year, \$100,000 mortgage with monthly payments. The bank is charging 1.5 points and “processing” costs of \$750; both points and processing costs are deducted from the mortgage when it is given. Monthly payments on the mortgage are based on the 12% annual interest rate on the full amount of the mortgage (that is, \$100,000).
  - a. Calculate the monthly payment on the mortgage, show the amortization table, and compute the EAIR.
  - b. Will the EAIR of the mortgage change if the loan period is 6 years?
  - c. Compute the total interest paid in each of the years of the mortgage. You can base your answer on the amortization table or investigate the Excel function **Cumipmt**.
7. You just bought the first floor of the famous “Egg-Plant” Building for \$250,000. You plan to rent the space to convenience stores. Your banker has offered you a mortgage with the following terms:
  - The mortgage is for the full amount of \$250,000.
  - The mortgage will be repaid in equal monthly payments over 36 months, starting 1 month from now.
  - The annual interest rate on the mortgage is 8%, compounded monthly (meaning:  $\frac{8\%}{12} = 2/3\%$  per month).
  - You have to pay the bank an initiation charge of \$1,500 and one point.
    - a. What is the monthly payment you will pay the bank?
    - b. What is the EAIR of the mortgage?
    - c. Compute an amortization table that shows the amount of interest you can report for taxes each month.
8. Your local bank has offered you a 5-year, \$100,000 mortgage. The bank is charging 1.5 points and “processing” costs of \$750. If the interest rate is 12% compounded monthly (meaning: 1% per month):
  - a. Calculate the monthly mortgage payment.
  - b. Calculate the EAIR.
  - c. Compute an amortization table that shows the amount of interest you can report for taxes each month and use it to compute the annual mortgage interest you can report for tax purposes.
9. You’re considering buying an asset that has a 3-year life and costs \$15,000. As an alternative to buying the asset, you can lease it for \$4,000 per year (four annual payments, the first due on the day you sign the lease). If you can borrow from your bank at 10%, should you lease or buy?
10. You’re considering buying an asset that has a 3-year life and costs \$2,000. As an alternative to buying the asset, you can lease it for \$600 per year (four annual payments, the first due today). Your bank is willing to lend you money for 15%.
  - a. Should you lease or purchase the asset?
  - b. What is the largest lease payment you would be willing to make?

## CHAPTER 3 What Does It Cost? IRR and the Time Value of Money 99

11. You're considering leasing or purchasing an asset with the following cash flows.
- Calculate the present value of the lease versus the purchase. Which is preferable.
  - What is the largest annual lease payment you would be willing to pay?

	A	B	C	D
1	<b>LEASE VERSUS PURCHASE WITH RESIDUAL VALUE</b>			
2	Asset cost	20,000		
3	Annual lease payment	5,500		
4	Residual value, year 3	3,000	<-- Value of asset at end year 3	
5	Bank rate	15%		
6				
7	<b>Year</b>	<b>Purchase cash flow</b>	<b>Lease cash flow</b>	
8	0	20,000	5,500	
9	1		5,500	
10	2		5,500	
11	3	-3,000	5,500	

12. You intend to buy a new laptop computer. Its price at electronic shops is \$2000, but your next-door neighbor offered to lease you the same computer for monthly payments of \$70 for a 24-month period, with the first payment made today. Assuming you can sell the computer at the end of 2 years at \$500 and the interest rate in the market is 20%, should you buy or lease?
13. Car leasing. You're considering leasing or purchasing a car. The details of each method of financing are given below. The lease is for 24 months. What should you do?

	A	B	C	D	E	F
1	<b>AUTO LEASE VERSUS PURCHASE</b>					
2	MSRP	50,000.00	<-- Manufacturer's suggested retail price			
3	Capitalized cost	45,000.00	<-- Negotiated price			
4	Destination charge	415.00	<-- Paid both by the lessee and the buyer			
5	Acquisition fee	450.00	<-- Paid only by the lessee			
6	Security deposit	450.00	<-- refunded at end of lease			
7						
8	Payment due at signing	1,315.00				
9	Monthly payment	600.00	<-- Dealer's lease offer			
10						
11	Residual value after 2 years as % of MSRP	60%				
12	Lease residual value after 3 years	30,000.00	<-- =B11*B2			
13	Your estimated residual value	35,000.00	<-- Your guess			
14						
15	Bank loan cost (annual)	7.00%				

14. Car leasing: You are considering buying a new, very expensive, FancyCar. The details of your negotiation for a 48-month lease are given below.
- If your alternative cost of financing is 6%, should you buy or lease?
  - If you financed the purchase through the bank at 6%, what would be your monthly loan payment?
  - As your estimated residual value (cell B15) gets larger, does leasing or buying become more advantageous? Explain.



## 100 PART ONE CAPITAL BUDGETING AND VALUATION

- d. What is the estimated residual value for which you are indifferent between buying and leasing?

	A	B	C
1	<b>AUTO LEASE VERSUS PURCHASE: 48 MONTH LEASE</b>		
2	MSRP	50,000	<-- Manufacturer's suggested retail price
3	Capitalized cost	45,000	<-- Negotiated price
4	Destination charge	415	<-- Paid both by the lessee and the buyer
5	Acquisition fee	450	<-- Paid only by the lessee
6	Security deposit	450	<-- refunded at end of lease
7			
8	Payment due at signing	1,315	
9	Monthly payment	400	<-- Dealer's lease offer
10			
11			
12	Residual value after 4 years as % of MSRP	60%	
13	Lease residual value after 4 years	30,000	
14	Your estimated residual value	35,000	<-- Your guess

15. You're considering buying a new top-of-the-line luxury car. The car's list price is \$99,000. The dealer has offered you two alternatives for purchasing the car:
- You can buy the car for \$90,000 in cash and get a \$9,000 discount in the bargain.
  - You can buy the car for the list price of \$99,000. In this case the dealer is willing to take \$39,000 as an initial payment. The remainder of \$60,000 is a "zero-interest loan" to be paid back in equal installments over 36 months.
- Alternatively, your local bank is willing to give you a car loan at an annual interest rate of 10%, compounded monthly (that is 10%/12 per month).
- Decide how to finance the car: Bank loan, zero-interest loan with the dealer, or cash payment.
16. You've been offered two credit cards:
- Credit card 1 charges 19% annually, on a monthly basis.
  - Credit card 2 charges 19% annually, on a weekly basis.
  - Credit card 3 charges 18.90% annually, on a daily basis.
- Rank the cards on the basis of EAIR.
17. You plan to put \$1,000 in a savings plan and leave it there for 5 years. You can choose between various alternatives. How much will you have in 5 years under each alternative?
- a. Bellon Bank is offering 12% stated annual interest rate, compounded once a year.
  - b. WNC Bank is offering 11% stated annual interest rate, compounded twice a year.
  - c. Plebian Bank is offering 10% stated annual interest rate, compounded monthly.
  - d. Byfus Bank is offering 11.5% stated annual interest rate, compounded continuously.
18. Assuming that the interest rate is 5%, compounded semiannually, which of the following is more valuable?
- a. \$5,000 today.
  - b. \$10,000 at the end of 5 years.
  - c. \$9,000 at the end of 4 years.
  - d. \$450 at the end of each year (in perpetuity) commencing in 1 year.
19. You plan to put \$10,000 in a savings plan for 2 years. How much will you have at the end of 2 years with each of the following options?

## CHAPTER 3 What Does It Cost? IRR and the Time Value of Money 101

- a. Receive 12% stated annual interest rate, compounded monthly.
- b. Receive 12.5% stated annual interest rate, compounded annually.
- c. Receive 11.5% stated annual interest rate, compounded daily.
- d. Receive 10% stated annual interest rate in the first year and 15% stated annual interest rate in the second year, compounded annually.

How much will you have at the end of 2 years in each option?

20. Michael Smith was in trouble: He was unemployed and living on his monthly disability pay of \$1,200. His credit card debts of \$19,000 were threatening to overwhelm this puny income. Every month in which he delayed paying the credit card debt cost him 1.5% on the remaining balance. His only asset was his house, on which he had a \$67,000 mortgage.

Then a Michael got a phone call from Uranus Financial Corporation: The company offered to refinance Michael's mortgage. The Uranus representative explained to Michael that, with the rise in real-estate values, Michael's house could now be remortgaged for \$90,000. This amount would allow Michael to repay his credit card debts and even leave him with some money.

Here are some additional facts:

- The new mortgage would be for 25 years and would have an annual interest rate of 9.23%. The mortgage would be repayable in equal monthly payments over this term, at a monthly interest rate of  $9.23\% / 12 = 0.76917\%$ . The fees on the mortgage are \$8,000.
- There are no penalties involved in repaying the \$67,000 existing mortgage.

Answer the following questions:

- a. What will Michael's monthly payments be on the new mortgage?
  - b. After repaying his credit card debts, how much money will Michael have left?
  - c. What is the effective annual interest rate (EAIR) on the Uranus mortgage?
21. A recent note from the author's credit card company included the following statement.
- Annual Percentage Rate for Cash Advances:**
- Your annual percentage rate for cash advances is the U.S. Prime Rate plus 14.99%, but such cash advance rate will never be lower than 19.99%. As of August 1, 2004, this cash advance ANNUAL PERCENTAGE RATE is 19.99%, which corresponds to a daily periodic rate of 0.0548%. A daily periodic rate is the applicable annual percentage rate divided by 365.
- As of August 1, 2004, what is the effective annual interest rate (EAIR) charged by the credit card on cash advances?
22. WindyRoad is an investment company that has two mutual funds. The WindyRoad Dull Fund invests in boring corporate bonds and its Lively Fund invests in "high-risk, high-return" companies. The returns for the two funds in the 5-year period 2001–2005 are given below.
- a. Suppose you had invested \$100 in each of the two funds at the beginning of 2001. How much would you have at the end of 2005?
  - b. What was the EAIR paid by each of the funds over the 5-year period 2001–2005?
  - c. Is there a conclusion you can draw from this example?

## 102 PART ONE CAPITAL BUDGETING AND VALUATION

	A	B	C	D
1	<b>DULL FUND OR LIVELY FUND?</b>			
2	<b>Year</b>	<b>Dull Fund return</b>	<b>Lively Fund return</b>	
3	2001	9.20%	11.50%	
4	2002	5.20%	-14.50%	
5	2003	4.30%	-23.40%	
6	2004	3.30%	42.40%	
7	2005	7.00%	13.60%	
8				
9	Average return	5.80%	5.92%	<-- =AVERAGE(C3:C7)

23. (Advanced).
- Compute the annual continuous returns for Dull Fund and Lively Fund (exercise 22) for each of the years 2001–2005. What is the average continuous return  $r_{\text{average}}^{\text{continuous}}$  for each fund?
  - Suppose you had invested \$100 in each of the two funds at the beginning of 2001. Show that the total amount you would have in each fund (see Exercise 22.a.) can be written as  $\$100 * e^{5 * r_{\text{average}}^{\text{continuous}}}$ . Note that this makes computations much simpler.
24. Capital Star Motors of Canberra, Australia, has the following lease offer for a Smart car:
- Lease term: 48 months.
  - No initial deposit, \$8,995 balloon payment at end of lease.
  - Daily payment: \$9.95 per day, payable monthly.
  - Smart car cash cost: \$18,800.
- Assuming 30 days per month, compute the effective annual interest rate (EAIR) in the lease.
  - Compute the balloon payment that gives a 7% EAIR.
25. Ten years ago Reem and Forsan took out a \$150,000 mortgage to buy their new house. The mortgage was a 20-year, 10% mortgage with annual payments. Today their bank offers similar mortgages at 8% interest; however, when the young couple went to the bank to inquire about the possibility of refinancing the remainder of their mortgage with a 10-year, 8% mortgage, they were informed that the bank has a \$15,000 “exit fee” for the refinancing.<sup>16</sup> Should they still refinance the mortgage?
- We offer some hints:
- The remaining mortgage principal at any point is the present value of the future payments on the mortgage. Further hint: use Excel’s PV function.
  - Excel’s Rate function can compute the IRRs of fixed payment loans.

<sup>16</sup>This question was inspired by an article in the *Melbourne Age* on 12 August 2009. High exit fees are common in Australia; they are also called “early termination fees,” “deferred establishment fees,” or “early repayment fees.”

## CHAPTER 3 What Does It Cost? IRR and the Time Value of Money 103

- A convenient template to solve this problem might be the following:

	A	B	C
1	<b>MORTGAGE EXIT FEE</b>		
2	<b>Initial mortgage</b>		
3	Principal	150,000.00	
4	Term (years)	20	
5	Interest rate	10.00%	
6	Annual payment		<--
7	Term remaining	10	
8	Remaining mortgage principal		<--
9	Exit fee	15,000.00	
10			
11	<b>New mortgage</b>		
12	Principal		<--
13	Term (years)	10	
14	Interest rate	8.00%	
15	Annual payment		<--
16	Effective annual cost (EAIR)		<--